
 

 

 

 
 

Application Note:  

Interfacing SolarSIM-D2 and SolarSIM-G with 

Campbell Scientific Dataloggers 

1. Introduction 
  

This application note describes how to interface the SolarSIM-D2 and the SolarSIM-G with a 

Campbell Scientific CR6 series datalogger. Three cases are considered: 

1. A single SolarSIM-G with a CR6 datalogger 

2. A single SolarSIM-D2 with a CR6 datalogger 

3. Both the SolarSIM-D2 and SolarSIM-G with a CR6 datalogger 

 

The provided instructions and programs can be easily adapted to other Campbell Scientific 

dataloggers. 

 

2. Wiring to a datalogger 
 

The SolarSIMs use the RS-485 communication protocol in a half-duplex mode. CR6 series 

datalogger, as shown in Figure 1, has two half-duplex RS-485 ports, which allows to acquire 

simultaneous data from both the SolarSIM-D2 and the SolarSIM-G.  

 

 
Figure 1. Interfacing with a Campbell Scientific’s CR6 series datalogger 

 

The wire colour guide for both the SolarSIM-D2 and the SolarSIM-G is as follows: 

 

Colour Label Function 

Blue Vin Input voltage (+12 VDC) 

White GND Common ground 

Black A- Negative RS-485 input 

Brown B+ Positive RS-485 input 

a. Wiring the SolarSIM-D2 to CR6 

SolarSIM-D2 Wire CR6 Connection Function 

Blue 12V 12V supply 

White G Common ground 

Black C1 Negative RS-485 input 

Brown C2 Positive RS-485 input 

b. Wiring the SolarSIM-G to CR6 

SolarSIM-G Wire CR6 Connection Function 

Blue 12V 12V supply 

White G Common ground 

Black C3 Negative RS-485 input 

Brown C4 Positive RS-485 input 

 

3. Datalogger communication 

a. Serial port configuration 
 

The serial port for the SolarSIM-D2 and the SolarSIM-G should be configured as follows:  

Parameter Value 

Baud rate 9600 

Parity None 

Data bits 8 

Stop bits 1 

 

b. Serial command for the SolarSIM-D2 

Only one serial command is needed to acquire raw data from the SolarSIM-D2: 

 

N100_E 
 

In return, the SolarSIM-D2 sends an ASCII string with the ambient pressure and temperature, the 

internal temperature and relative humidity, and six channel voltages. The following ASCII string is 

a sample output: 

 

N110_1013.120,2500.000,2600.000,1050.000,2500.032,4999.999, 

0000.001,1274.004,2746.321,3291.214/r/n 

This string can be parsed as: 

 

N”serial number”_ “Pout x 10” , ”(Tout + 50) x 75” , “(Tin + 50) × 75” , “Hin x 100” , “V1”, “V2” , “V3” , 

“V4” , “V5” , “V6”“end of line character” 

 

where: 

Pout =  ambient atmospheric pressure 

Tout =  ambient temperature 

Tin =  internal SolarSIM-D2 temperature 

Hin =  internal SolarSIM-D2 relative humidity 

 

The example string is further parsed in the table below: 

 

Parameter Symbol Value Units 

Ambient pressure Pout 101.312 kPa 

Ambient temperature Tout -16.67 ⷪC 

Internal temperature Tin -15.33 ⷪC 

Internal humidity Hin 10.50 % 

Voltage channel 1 V1 2500.032 mV 

Voltage channel 2 V2 4999.999 mV 

Voltage channel 3 V3 0.001 mV 

Voltage channel 4 V4 1274.004 mV 

Voltage channel 5 V5 2746.321 mV 

Voltage channel 6 V6 3291.214 mV 

 

c. Serial command for the SolarSIM-G 

 

Only one serial command is needed to acquire raw data from the SolarSIM-G: 

 

N1000_E 
 

In return, the SolarSIM-G sends an ASCII string with the ambient temperature, pressure and 

relative humidity, the internal temperature and relative humidity, and nine channel voltages. The 

following ASCII string is a sample output: 

 

N1010_2500.000,1013.120,4750.000,2600.000,1050.000,2500.032,4999.999, 

0000.001,1274.004,2746.321,3291.214, 3924.385,1900.500,0500.123/r/n 

 

This string can be parsed as: 

N”serial number”_”(Tout + 50) x 75” , “Pout x 10” , “Hout x 100” , “(Tin + 50) × 75” , “Hin x 100” , “V1”, 

“V2” , “V3” , “V4” , “V5” , “V6” , “V7” , “V8” , “V9”“end of line character” 

 

where: 

Tout =  ambient temperature 

Pout =  ambient atmospheric pressure 

Hout =  ambient relative humidity 

Tin =  internal SolarSIM-G temperature 

Hin =  internal SolarSIM-G relative humidity 

 

The aforementioned example string is parsed in the table below: 

 

Parameter Symbol Value Units 

Ambient temperature Tout -16.67 ⷪC 

Ambient pressure Pout 101.312 kPa 

Ambient humidity Hout 47.50 % 

Internal temperature Tin -15.33 ⷪC 

Internal humidity Hin 10.50 % 

Voltage channel 1 V1 2500.032 mV 

Voltage channel 2 V2 4999.999 mV 

Voltage channel 3 V3 0.001 mV 

Voltage channel 4 V4 1274.004 mV 

Voltage channel 5 V5 2746.321 mV 

Voltage channel 6 V6 3291.214 mV 

Voltage channel 7 V7 3924.385 mV 

Voltage channel 8 V8 1900.500 mV 

Voltage channel 9 V9 500.123 mV 

 

4. CRBasic Code 

a. For the SolarSIM-D2 
 
'Description: Acquires raw data from the SolarSIM-D2 
'Datalogger:  CR6 from Campbell Scientific 
'Author:      Spectrafy Inc. 
'Date:        February 20, 2020 
 
'Defines user constants 
Const Timezone     = -5 'hrs 
Const DAQRate      = 60 's  
Const SamplingRate = 5  's 

'Defines program constants  
Const TerminationCharacter = CHR(10) 
Const SerialCommand = "N100_E" 
 
'Declares public variables 
Public SerialData As String *256 
Public OutputData(10) 
 
'Declares data table column names 
Alias OutputData(1)  = AmbientPressure      'kPa   
Alias OutputData(2)  = AmbientTemperature   'C 
Alias OutputData(3)  = InternalTemperature  'C 
Alias OutputData(4)  = InternalHumidity     '% 
Alias OutputData(5)  = ChannelVoltage1      'mV 
Alias OutputData(6)  = ChannelVoltage2      'mV 
Alias OutputData(7)  = ChannelVoltage3      'mV 
Alias OutputData(8)  = ChannelVoltage4      'mV 
Alias OutputData(9)  = ChannelVoltage5      'mV 
Alias OutputData(10) = ChannelVoltage6      'mV 
 
'Defines data table 
DataTable (Spectrafy_D2,1,-1)       'Autoallocates table size                    
  DataInterval (0,DAQRate,Sec,10)   'Sets the DAQ rate 
  Sample(1,Timezone,FP2)            'Stores timezone 
  FieldNames("Timezone")            'Names "Timezone" column 
  Average(10, OutputData(),IEEE4,0) 'Stores raw data 
EndTable 
 
'Executes main program 
BeginProg 
  'Initializes serial port 
  SerialOpen (ComC1,9600,0,0,256,4)  'RS-485 communication on port "ComC1" 
                                     'Black wire (A-) to C1 terminal 
                                     'Brown wire (B+) to C2 terminal 
                                     'Baud rate: 9600 bps 
                                     'Buffer size: 256 bytes 
                                     'Mode: 4 (half-duplex RS485)             
  'Sets a 5 s scan interval 
  Scan (5,Sec,0,0)  
 
      'Transmits the broadcast command   
    SerialOut (ComC1,SerialCommand,"",0,0)  
      
    'Receives serial data with a 1000 ms timeout         
    SerialIn (SerialData,ComC1,100,TerminationCharacter,256)  
     
      'Clears the serial buffer 
    SerialFlush (ComC1)       
     
      'Removes the header from the serial data      

    SerialData = Mid(SerialData,6,256) 
     
      'Parses the serial data into numeric values 
    SplitStr (OutputData(),SerialData,"",10,0) 
     
      'Converts raw data into meteorological data 
      AmbientPressure = AmbientPressure / 10.0 
    AmbientTemperature = (AmbientTemperature / 75.0) - 50.0 
    InternalTemperature = (InternalTemperature / 75.0) - 50.0 
      InternalHumidity = InternalHumidity / 100.0 
       
      'Passes raw data to "Spectrafy" table 
    CallTable Spectrafy_D2 
  NextScan 
EndProg 
 

b. For the SolarSIM-G 
 
'Description: Acquires raw data from the SolarSIM-G 
'Datalogger:  CR6 from Campbell Scientific 
'Author:      Spectrafy Inc. 
'Date:        February 20, 2020 
 
'Defines user constants 
Const Timezone     = -5 'hrs 
Const DAQRate      = 60 's  
Const SamplingRate = 5  's 
 
'Defines program constants  
Const TerminationCharacter = CHR(10) 
Const SerialCommand = "N1000_E" 
 
'Declares public variables 
Public SerialData As String *256 
Public OutputData(14) 
 
'Declares data table column names 
Alias OutputData(1)  = AmbientTemperature   'C 
Alias OutputData(2)  = AmbientPressure      'kPa 
Alias OutputData(3)  = AmbientHumidity      '% 
Alias OutputData(4)  = InternalTemperature  'C 
Alias OutputData(5)  = InternalHumidity     '% 
Alias OutputData(6)  = ChannelVoltage1      'mV 
Alias OutputData(7)  = ChannelVoltage2      'mV 
Alias OutputData(8)  = ChannelVoltage3      'mV 
Alias OutputData(9)  = ChannelVoltage4      'mV 
Alias OutputData(10) = ChannelVoltage5      'mV 
Alias OutputData(11) = ChannelVoltage6      'mV 

Alias OutputData(12) = ChannelVoltage7      'mV 
Alias OutputData(13) = ChannelVoltage8      'mV 
Alias OutputData(14) = ChannelVoltage9      'mV 
 
'Defines data table 
DataTable (Spectrafy_G,1,-1)      'Autoallocates table size  
  DataInterval (0,DAQRate,Sec,10) 'Sets the DAQ rate 
  Sample (1,Timezone,FP2)         'Stores timezone 
  FieldNames("Timezone")          'Names "Timezone" column 
  Average (14,OutputData,IEEE4,0) 'Stores raw data 
EndTable 
 
'Executes main program 
BeginProg 
  'Initializes serial port 
  SerialOpen (ComC3,9600,0,0,256,4)  'RS-485 communication on port "ComC3" 
                                     'Black wire (A-) to C3 terminal 
                                     'Brown wire (B+) to C4 terminal 
                                     'Baud rate: 9600 bps 
                                     'Buffer size: 256 bytes 
                                     'Mode: 4 (half-duplex RS485)             
  'Sets a 5 s scan interval 
  Scan (SamplingRate,Sec,0,0)  
 
      'Transmits the broadcast command   
    SerialOut (ComC3,SerialCommand,"",0,0)  
      
    'Receives serial data with a 1000 ms timeout         
    SerialIn (SerialData,ComC3,100,TerminationCharacter,256)  
     
      'Clears the serial buffer 
    SerialFlush (ComC3)       
     
      'Removes the header from the serial data      
    SerialData = Mid(SerialData,7,256) 
     
      'Parses the serial data into numeric values 
    SplitStr (OutputData(),SerialData,"",14,0) 
     
      'Converts raw data into meteorological data 
       AmbientTemperature = (AmbientTemperature / 75.0) - 50.0 
       AmbientPressure = AmbientPressure / 10.0 
       AmbientHumidity = AmbientHumidity / 100.0 
       InternalTemperature = (InternalTemperature / 75.0) - 50.0 
       InternalHumidity = InternalHumidity / 100.0 
       
      'Passes raw data to "Spectrafy" table 
    CallTable Spectrafy_G 
  NextScan 
EndProg 

c. For the SolarSIM-D2 and the SolarSIM-G 
 
'Description: Acquires raw data from the SolarSIM-G 
'Datalogger:  CR6 from Campbell Scientific 
'Author:      Spectrafy Inc. 
'Date:        February 20, 2020 
 
'Defines user constants 
Const Timezone     = -5 'hrs 
Const DAQRate      = 60 's  
Const SamplingRate = 5  's 
 
'Defines program constants  
Const TerminationCharacter = CHR(10) 
Const SerialCommand_G =  "N1000_E" 
Const SerialCommand_D2 = "N100_E" 
 
'Declares public variables 
Public SerialData As String *256 
Public OutputData_G(14) 
Public OutputData_D2(10) 
 
'Declares data table column names for the SolarSIM-G 
Alias OutputData_G(1)  = AmbientTemperature_G     'C 
Alias OutputData_G(2)  = AmbientPressure_G        'kPa 
Alias OutputData_G(3)  = AmbientHumidity_G        '% 
Alias OutputData_G(4)  = InternalTemperature_G    'C 
Alias OutputData_G(5)  = InternalHumidity_G       '% 
Alias OutputData_G(6)  = ChannelVoltage1_G        'mV 
Alias OutputData_G(7)  = ChannelVoltage2_G        'mV 
Alias OutputData_G(8)  = ChannelVoltage3_G        'mV 
Alias OutputData_G(9)  = ChannelVoltage4_G        'mV 
Alias OutputData_G(10) = ChannelVoltage5_G        'mV 
Alias OutputData_G(11) = ChannelVoltage6_G        'mV 
Alias OutputData_G(12) = ChannelVoltage7_G        'mV 
Alias OutputData_G(13) = ChannelVoltage8_G        'mV 
Alias OutputData_G(14) = ChannelVoltage9_G        'mV 
 
'Declares data table column names for the SolarSIM-D2 
Alias OutputData_D2(1)  = AmbientPressure_D2      'kPa 
Alias OutputData_D2(2)  = AmbientTemperature_D2   'C   
Alias OutputData_D2(3)  = InternalTemperature_D2  'C 
Alias OutputData_D2(4)  = InternalHumidity_D2     '% 
Alias OutputData_D2(5)  = ChannelVoltage1_D2      'mV 
Alias OutputData_D2(6)  = ChannelVoltage2_D2      'mV 
Alias OutputData_D2(7)  = ChannelVoltage3_D2      'mV 
Alias OutputData_D2(8)  = ChannelVoltage4_D2      'mV 
Alias OutputData_D2(9)  = ChannelVoltage5_D2      'mV 
Alias OutputData_D2(10) = ChannelVoltage6_D2      'mV 

'Defines data table for SolarSIM-G 
DataTable (Spectrafy_G,1,-1)      'Autoallocates table size  
  DataInterval (0,DAQRate,Sec,10) 'Sets the DAQ rate 
  Sample (1,Timezone,FP2)         'Stores timezone 
  FieldNames("Timezone")          'Names "Timezone" column 
  Average (14,OutputData_G,IEEE4,0) 'Stores raw data 
EndTable 
 
'Defines data table for SolarSIM-D2 
DataTable (Spectrafy_D2,1,-1)     'Autoallocates table size  
  DataInterval (0,DAQRate,Sec,10) 'Sets the DAQ rate 
  Sample (1,Timezone,FP2)         'Stores timezone 
  FieldNames("Timezone")          'Names "Timezone" column 
  Average (10,OutputData_D2,IEEE4,0) 'Stores raw data 
EndTable 
 
'Executes main program 
BeginProg 
  'Initializes serial port for the SolarSIM-G 
  SerialOpen (ComC3,9600,0,0,256,4)  'RS-485 communication on port "ComC3" 
                                     'Black wire (A-) to C3 terminal 
                                     'Brown wire (B+) to C4 terminal 
                                     'Baud rate: 9600 bps 
                                     'Buffer size: 256 bytes 
                                     'Mode: 4 (half-duplex RS485) 
                                      
  'Initializes serial port for the SolarSIM-D2 
  SerialOpen (ComC1,9600,0,0,256,4)  'RS-485 communication on port "ComC1" 
                                     'Black wire (A-) to C3 terminal 
                                     'Brown wire (B+) to C4 terminal 
                                     'Baud rate: 9600 bps 
                                     'Buffer size: 256 bytes 
                                     'Mode: 4 (half-duplex RS485)                                                                                   
 
'-----------------------------SolarSIM-G--------------------------------- 
  'Sets a 5 s scan interval 
  Scan (SamplingRate,Sec,0,0)         
 
      'Transmits the broadcast command   
    SerialOut (ComC3,SerialCommand_G,"",0,0)  
      
    'Receives serial data with a 1000 ms timeout         
    SerialIn (SerialData,ComC3,100,TerminationCharacter,256)  
     
      'Clears the serial buffer 
    SerialFlush (ComC3)       
     
      'Removes the header from the serial data      
    SerialData = Mid(SerialData,7,256) 

      'Parses the serial data into numeric values 
    SplitStr (OutputData_G(),SerialData,"",14,0) 
     
      'Converts raw data into meteorological data 
       AmbientTemperature_G = (AmbientTemperature_G / 75.0) - 50.0 
       AmbientPressure_G = AmbientPressure_G / 10.0 
       AmbientHumidity_G = AmbientHumidity_G / 100.0 
       InternalTemperature_G = (InternalTemperature_G / 75.0) - 50.0 
       InternalHumidity_G = InternalHumidity_G / 100.0 
       
      'Passes raw SolarSIM-G data to "Spectrafy" table 
     CallTable Spectrafy_G 
      
'----------------------------SolarSIM-D2---------------------------------       
      'Transmits the broadcast command   
    SerialOut (ComC1,SerialCommand_D2,"",0,0)  
      
    'Receives serial data with a 1000 ms timeout         
    SerialIn (SerialData,ComC1,100,TerminationCharacter,256)  
     
      'Clears the serial buffer 
    SerialFlush (ComC1)       
     
      'Removes the header from the serial data      
    SerialData = Mid(SerialData,6,256) 
     
      'Parses the serial data into numeric values 
    SplitStr (OutputData_D2(),SerialData,"",10,0) 
     
      'Converts raw data into meteorological data 
      AmbientPressure_D2 = AmbientPressure_D2 / 10.0 
    AmbientTemperature_D2 = (AmbientTemperature_D2 / 75.0) - 50.0 
    InternalTemperature_D2 = (InternalTemperature_D2 / 75.0) - 50.0 
      InternalHumidity_D2 = InternalHumidity_D2 / 100.0 
 
     'Passes raw SolarSIM-D2 data to "Spectrafy" table 
     CallTable Spectrafy_D2 
 NextScan 
  
EndProg 

5. Support 
 

If you have any questions regarding your specific application, don’t hesitate to contact Spectrafy 

at info@spectrafy.com. 
 


